
*=============A_PM_Jitter_Plot==============================

It is possible to do a jitter plot by doing
a little processing of the output waveform.

First off, Phase Modulation at +/- 1 radian will be applied
at a 1Hz rate. 

Period_Jitter_Plot
* dsauersanjose@aol.com  2.10.10_3.12PM
* www.idea2ic.com                 replace(OPT-SPACE)=>SPACE
*    ___
*   |OUT|   .5sidebands should product +/- 1.0radians PM
*   |___|    +/- 1radian/6.28318 = +/- 16%
*    _|_     
*   /_  \
*  // \  \
*  \   \//
*   \___/
*    _|_
*    /// 
.OPTIONS  METHOD     =gear   
*=========The_SideBand_Tests=========================
VT        VT     0      PWL(   0      0      2       2 )   DC     0
VP        VP     0      DC     3.141592653589793
V2P       V2P    0      DC     6.283185307179586
BPM1      VPM1   0      V =    cos( v(V2P)*10*v(VT) +1*cos( v(V2P)*1*v(VT)) )
*================================================================

It may be easier to define PM as an equation.

*=========Clipping_Circuit=========================
BPM1C     VPM1C  0      V =    2*u( v(VPM1))-1
*=========Run_Sim=========================
.control 
set       pensize = 2 
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .1m     2      0       .1m
*============PlotResults============================================
plot      vpm1    
plot      vpm1c   
*================================================================

The carrier is phase modulated at +/-1 radians and then
clipped. 

The output Array can be processed to find rising and falling
edges.

Assume the number of rising or falling edges are not known
at this point. So array anysize will be used to store
an unknown number of data points. The total number of output 
points (num) for the oscillator output is easy to find.

Some simple "if" statements can be used to find the timing 
for the edges. 

*=========Create_AnySize_Arrays============================================
compose   anysize     start = 0 stop = 99 step =1
let       num =       length(vpm1c)-5
let i  =  0
let t  =  0
let n  =  0
*=========Find_Edge_Timing============================================
repeat    $&num 
if        ( vpm1[i] < 0 & vpm1[i+1] > 0)
let t =   time[i]
let       anysize[n]= t



echo      n=   $&n   out_rise= $&t
let       n =   n +1    
endif
if        ( vpm1[i] > 0 & vpm1[i+1] < 0)
let       t =   time[i]
let       anysize[n]= t
echo      n=   $&n   out_fall= $&t
let       n =   n +1    
endif
let       i =   i +1
endrepeat
let       n3 =  n -1
*================================================================

The MacSpice printout is as follows.

Circuit: Period_Jitter_Plot

n = 0  out_fall = 0.0091024
n = 1  out_rise = 0.0602024
n = 2  out_fall = 0.112902
n = 3  out_rise = 0.167002
n = 4  out_fall = 0.222202
n = 5  out_rise = 0.277702
n = 6  out_fall = 0.332902
n = 7  out_rise = 0.387002
n = 8  out_fall = 0.439702
n = 9  out_rise = 0.490802
n = 10 out_fall = 0.540402
n = 11 out_rise = 0.588502
n = 12 out_fall = 0.635402
n = 13 out_rise = 0.681602
n = 14 out_fall = 0.727202
n = 15 out_rise = 0.772702
n = 16 out_fall = 0.818302
n = 17 out_rise = 0.864502
n = 18 out_fall = 0.911402
n = 19 out_rise = 0.959502
n = 20 out_fall = 1.0091
n = 21 out_rise = 1.0602
n = 22 out_fall = 1.1129
n = 23 out_rise = 1.167
n = 24 out_fall = 1.2222
n = 25 out_rise = 1.2777
n = 26 out_fall = 1.3329
n = 27 out_rise = 1.387
n = 28 out_fall = 1.4397
n = 29 out_rise = 1.4908
n = 30 out_fall = 1.5404
n = 31 out_rise = 1.5885
n = 32 out_fall = 1.6354
n = 33 out_rise = 1.6816
n = 34 out_fall = 1.7272
n = 35 out_rise = 1.7727
n = 36 out_fall = 1.8183
n = 37 out_rise = 1.8645
n = 38 out_fall = 1.9114
n = 39 out_rise = 1.9595

*================================================================

Now that the number of edge data points are known,
some new arrays can be created to store and plot
the results.

This further can get processed to find time periods
which can then be plotted.

*=========Create_Edge_Time_Arrays============================================

compose   tp           start = 0 stop = $&n3 step =1
compose   tpac         start = 0 stop = $&n3 step =1
compose   td           start = 0 stop = $&n3 step =1
compose   tdac         start = 0 stop = $&n3 step =1
compose   rtp          start = 0 stop = $&n3 step =1
*=========Transfer_Arrays============================================
let       i  =  0
repeat    $&n
let       rtp[i] = anysize[i]
let       i =   i +1
endrepeat 
let       i  =  0



let       n2 =  n -1
repeat    $&n2
let       tp[i] = rtp[i+1] -rtp[i]
let       i =   i +1
endrepeat 
let       tp[n2] = tp[n2-1]
plot      td    vs   rtp
*================================================================

In this case rtp stands for time reference point.
That is the time between when the transitions happen.
The value tp stands for time period. This is
the actual time between edges.

The edge to edge period for a 10Hz square wave is
50msec on the average.

It is easy to do some further math on the data.

*=========Remove_Average_Time_Period============================================
let       tpave = mean(tp)
let       tpac =  tp -tpave
plot      tpac vs rtp
*================================================================
The average time period can be subtracted out. 

Note, modulating the time period is in effect 
Frequency modulation. 

*=========Convert_FM_to_PM============================================
let i  =  1
let n2 =  n -1
repeat    $&n2
let       td[i] = td[i-1] +tpac[i]
let i =   i +1



endrepeat
plot      td    vs   rtp
*================================================================

To see the resulting phase shift in terms of time,
all the time periods need to be added up.

*=========Remove_Average_Phase============================================
let       tdave = mean(td)
let       tdac  =  td -tdave
plot      td    vs rtp
plot      tdac  vs rtp
*================================================================
.endc 
.end 
*================================================================

Here rtp stands for a reference time point.
This is the time when either a rising or falling
edge happens.

The delay tdac represents the time between the
actual edge compared to a perfect 10Hz square wave.

Math says that the delay should be about +/- 1 radians
which should be about +/- 16msec. The plot shows 
about the same number.

Period_Jitter_Plot
* dsauersanjose@aol.com  2.15.10_4.38PM
* www.idea2ic.com        replace(OPT-SPACE)=>SPACE
*    ___
*   |OUT|    +/- 1radian/6.28318 = +/- 16% of 100msec
*   |___|    
*    _|_     



*   /_  \
*  // \  \
*  \   \//
*   \___/
*    _|_
*    /// 

===========Full_Netlist_For_Copy_Paste=======================
Period_Jitter_Plot
* dsauersanjose@aol.com  2.15.10_4.42PM
* www.idea2ic.com                 replace(OPT-SPACE)=>SPACE
*    ___
*   |OUT|   .5sidebands should product +/- 1.0radians PM
*   |___|    +/- 1radian/6.28318 = +/- 16%
*    _|_     
*   /_  \
*  // \  \
*  \   \//
*   \___/
*    _|_
*    ///*
.OPTIONS METHOD     =gear   
*=========The_SideBand_Tests=========================
VT        VT     0      PWL(   0      0      2       2 )   DC     0
VP        VP     0      DC     3.141592653589793
V2P       V2P    0      DC     6.283185307179586
BPM1      VPM1   0      V =    cos( v(V2P)*10*v(VT) +cos( v(V2P)*1*v(VT)) )
*=========Clipping_Circuit=========================
BPM1C     VPM1C  0      V =    2*u( v(VPM1))-1
*=========Run_Sim=========================
.control 
set       pensize = 2 
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .1m     2      0       .1m
*============PlotResults============================================
plot      vpm1    
plot      vpm1c   
*=========Create_AnySize_Arrays============================================
compose   anysize     start = 0 stop = 99 step =1
let       num =       length(vpm1c)-5
let i  =  0
let t  =  0
let n  =  0
*=========Find_Edge_Timing============================================
repeat    $&num 
if        ( vpm1[i] < 0 & vpm1[i+1] > 0)
let t =   time[i]
let       anysize[n]= t
echo      n=   $&n   out_rise= $&t
let       n =   n +1    
endif
if        ( vpm1[i] > 0 & vpm1[i+1] < 0)
let t =   time[i]
let       anysize[n]= t
echo      n=   $&n   out_fall= $&t
let       n =   n +1    
endif
let i =   i +1
endrepeat
let       n3 =  n -1
*=========Create_Edge_Time_Arrays============================================
compose   tp           start = 0 stop = $&n3 step =1
compose   tpac         start = 0 stop = $&n3 step =1
compose   td           start = 0 stop = $&n3 step =1
compose   tdac         start = 0 stop = $&n3 step =1
compose   rtp          start = 0 stop = $&n3 step =1
*=========Transfer_Arrays============================================
let i  =  0
repeat    $&n
let       rtp[i] = anysize[i]
let i =   i +1
endrepeat 
let i  =  0
let n2 =  n -1
repeat    $&n2
let       tp[i] = rtp[i+1] -rtp[i]
let i =   i +1
endrepeat 
let       tp[n2] = tp[n2-1]
plot      tp  vs rtp
*=========Remove_Average_Time_Period============================================
let       tpave = mean(tp)



let       tpac =  tp -tpave
plot      tpac vs rtp
*=========Convert_FM_to_PM============================================
let i  =  1
let n2 =  n -1
repeat    $&n2
let       td[i] = td[i-1] +tpac[i]
let i =   i +1
endrepeat 
plot      td    vs rtp
*=========Remove_Average_Phase============================================
let       tdave = mean(td)
let       tdac  =  td -tdave
plot      tdac  vs rtp

.endc 

.end

2.12.10_2.55PM 

